10 - “And” Goal

Lean: First Steps

Tariq Rashid

September 29, 2024

“And” Goal

- We’ve seen disjunctive and conjunctive **hypotheses**.
- We’ve also seen disjunctive proof **goals**.
- Here we look at a conjunctive proof **goal**.

Task

- For integer x, given that $x = -1$, show that

$$(x^2 = 1) \wedge (x^3 = -1)$$

- $P \wedge Q$ true means at **both** P and Q must be true.
- So we have to **prove** both $x^2 = 1$ and $x^3 = -1$.

Maths

$$x = -1 \quad \text{given fact} \quad (1)$$

$$(x^2 = 1) \wedge (x^3 = -1) \quad \text{proof objective}$$

$$\begin{aligned} x^2 &= (-1)^2 && \text{using fact (1)} \\ &= 1 && \text{using arithmetic} \quad (2) \end{aligned}$$

$$\begin{aligned} x^3 &= (-1)^3 && \text{using fact (1)} \\ &= -1 && \text{using arithmetic} \quad (3) \end{aligned}$$

$$(x = -1) \implies (x^2 = 1) \wedge (x^3 = -1) \quad \text{using results (2, 3)} \quad \square$$

Code

```
-- 10 - Conjunctive "and" Goal

import Mathlib.Tactic

example {x : ℤ} (h : x = -1) : x^2 = 1 ∧ x^3 = -1 := by
  constructor
  · calc
      x^2 = (-1)^2 := by rw [h]
      _ = 1 := by norm_num
  · calc
      x^3 = (-1)^3 := by rw [h]
      _ = -1 := by norm_num
```

Code

- The objective here is a conjunction, uses the symbol $\wedge$ to denote "logical and".
- `constructor` splits the conjunctive goal into two separate goals.
- Infoview will confirm `constructor` replaces the **single** goal `x^2 = 1 ∧ x^3 = 1` with **two** new goals, `x^2 = 1` and `x^3 = -1`.
- The rest of the proof has two **calc** structures, one after the other, to show `x^2 = 1` and `x^3 = -1`.
  - Focussing dots are used to make clear there are two sub-proofs.

Infoview

- Placing the cursor before `constructor` shows the original proof goal.

```
x : ℤ
h : x = -1
⊢ x ^ 2 = 1 ∧ x ^ 3 = -1
```

- The cursor after `constructor` shows the two replacement goals.

```
x : ℤ
h : x = -1
⊢ x ^ 2 = 1

x : ℤ
h : x = -1
⊢ x ^ 3 = -1
```

Easy Exercise

- For integer x, given that $x = -1$, write a Lean program to show that

$$(x^3 = -1) \wedge (x^4 = 1) \wedge (x^5 = -1)$$

- A longer conjunction could be $P \wedge Q \wedge R \wedge S$.
  - `constructor` splits this into two goals, P and $Q \wedge R \wedge S$.
- The proof will require two uses of `constructor`.